# Integral Bounds and RegionPlot3D

We a tough problem in class today preparing for the final exam next Monday.

$$\int_0^{\pi^{1/4}}\int_0^z\int_y^z 12y^2z^3\sin(x^4)\,dx\,dy\,dz$$

I was extremely surprised that I could simply type in the bounds of the integrals as inequalities using the RegionPlot3D command.

RegionPlot3D[0 <= z <= π^(1/4) && 0 <= y <= z && y <= x <= z, {x, 0, π^(1/4)},
{y, 0, π^(1/4)}, {z, 0, π^(1/4)}]


Which produced the correction region of integration.

Will this process with the bounds work for every such integral? If not, can someone provide a counterexample?

• I am surprised by your surprise. What did you think the RegionPlot3D function was for if not this? – user484 Dec 4 '15 at 22:28
• I'd say the only time things will mess up is when your integral limits are flipped. i.e. if you're integrating z from pi^(1/4) to 0, then your inequality is always false. – Chip Hurst Dec 4 '15 at 22:32
• f[x_, y_, z_] := 12 y^2 z^3 Sin[x^4]; Integrate[ f[x, y, z] Boole[0 <= z <= \[Pi]^(1/4) && 0 <= y <= z && y <= x <= z ], {x, 0, \[Pi]^(1/4)}, {y, 0, \[Pi]^(1/4)}, {z, 0, \[Pi]^(1/4)}] is more surprising :) – Dr. belisarius Dec 4 '15 at 23:39

rgn = ImplicitRegion[
0 < z < π^(1/4) && 0 < y < z &&
y < x < z, {x, y, z}];

(int1 = Integrate[
12 y^2 z^3 Sin[x^4],
{x, y, z} ∈ rgn]) // AbsoluteTiming

(*  {1.38881, π/4}  *)

(int2 = Assuming[{Element[{x, y, z}, Reals]},
Integrate[12 y^2 z^3 Sin[x^4],
{z, 0, π^(1/4)}, {y, 0, z},
{x, y, z}]]) // AbsoluteTiming

(*  {7.53664, π/4}  *)

int1 == int2

(*  True  *)

RegionPlot3D[rgn,
PlotPoints -> 100,
Axes -> True,
AxesLabel ->
(Style[#, 14, Bold] & /@ {"x", "y", "z"})]


• Interesting that the integration over an implicit region is faster. – David Dec 5 '15 at 8:50